

# MODELADO DE TRANSISTORES PHEMT DE MICROONDAS EN SPICE

J. M. Zamanillo, H. Ingelmo, R. Toyos, C. Pérez-Vega, A. Mediavilla

Departamento de Ingeniería de Comunicaciones

Grupo de RF & Microondas

Universidad de Cantabria

e-mail : [jose.zamanillo@unican.es](mailto:jose.zamanillo@unican.es)

**Abstract-** As an extension of our previous works in the electrical modelling of microwave, this paper shows the result of the research on large signal behaviour (DC I/V curves) of AlGaAs P-HEMT (pseudomorphic high electron mobility transistor) devices, in the overall I/V plane, and how is possible to include the model into the PSPICE simulator. Experimental results show very good agreement with the theoretical analysis.

## I. INTRODUCCIÓN

En los últimos tiempos nuestro grupo de investigación ha desarrollado números modelos de dispositivos MESFET y HEMT, incluso bajo iluminación óptica, los cuales ofrecen unos resultados totalmente en concordancia con las medidas experimentales [1,3]. Los modelos desarrollados solamente habían sido introducidos en simuladores basados en el balance armónico como el simulador ADS Agilent Technologies. Dicho simulador, es a nuestro parecer, uno de los más completos y eficientes existentes en el mercado, pero presenta el inconveniente de su elevado precio, el cual solamente se encuentra al alcance de ciertas grandes empresas y de las universidades debido a la política de descuento por licencias educacionales, seguida por Agilent. Con la idea de popularizar los modelos desarrollados en el seno de nuestro grupo de modelado, surgió la pregunta siguiente: *¿Por qué no utilizar SPICE como simulador de dispositivos de microondas?* SPICE es una herramienta universalmente conocida y probada, cuyo precio es muy razonable, y además existen numerosas licencias de estudiante las cuales pueden obtenerse de forma gratuita de la mayoría de fabricantes del producto como ORCAD o Intusoft.




Fig 1. Modelo DICOM de gran señal para HEMT/MESFET.

La implementación del modelo en el simulador PSPICE (versión de SPICE para PC) se ha dividido en dos partes: en la primera, la cual es el objeto de la presente comunicación se mostrará como ha sido introducido el modelo de PHEMT de microondas en SPICE y se validaran las curvas características I-V obtenidas mediante el simulador con medidas experimentales realizadas en nuestro laboratorio. En una segunda fase, en la que se trabaja en la actualidad, y cuyos resultados se mostraran en futuras comunicaciones se mostrará el comportamiento del modelo en cuanto a parámetros de scattering validándose en su día con medidas realizadas en el laboratorio.

## II. EL SIMULADOR

El simulador SPICE fue desarrollado por la Universidad de Berkley en la década de los 70' del pasado siglo, y aunque hoy en día existen numerosos simuladores, es uno de los de mayor difusión en el seno de la comunidad científica.

Todos los modelos de dispositivos activos incluidos en SPICE, incluyendo el de MESFET de AsGa son de origen físico [4] y por tanto los parámetros que los definen dependen de la tecnología de fabricación del transistor, así como de factores geométricos solamente conocidos por el fabricante o el diseñador del dispositivo. En esta comunicación se quiere dar un paso más allá e introducir un modelo eléctrico basado en el concepto de circuito equivalente que además de cuenta de la potencia óptica aplicada al transistor. SPICE es un simulador basado en el análisis de los estímulos y respuestas del circuito a analizar en el dominio del tiempo, aunque permite mediante transformada de Fourier pasar al dominio de la frecuencia. Dado los modelos de transistores desarrollados en nuestro grupo se basan en la frecuencia como variable de barrido, en la presente comunicación solamente se introducirá en el simulador la parte del modelo que reproduce las curvas características tensión-corriente (I-V) de continua, dejándose el resto del modelo (curvas pulsadas y parámetros de scattering) como objetivo para posteriores trabajos.

## III. EL MODELO

Entre los diferentes modelos clásicos existentes en la bibliografía, se eligió el modelo DICOM [1] desarrollado con anterioridad por nuestro grupo por ser el adecuado para representar con fidelidad el comportamiento de dichos dispositivos. El esquema eléctrico del modelo de gran señal propuesto se muestra en la figura 1, dicho modelo es valido

tanto para transistores MESFET como para transistores HEMT. Nótese que por tratarse de un modelo para DC se han omitido las inducciones parásitas Lg, Ld, Ls, así como las capacidades parásitas Cpgi y Cpdi.

Los modelos clásicos tales como el de Curtice, Materka, TOM “*Trquint Own Model*”, etc., ofrecen, en general, buenos resultados en transistores de baja potencia, sin embargo no son capaces de responder de manera adecuada para dispositivos HEMT de potencia. El modelo DICOM que ha sido seleccionado además de suplir esta carencia de los modelos precedentes es capaz de mostrar un fenómeno característico de los transistores HEMT: el fenómeno de compresión de trasconductancia a medida que aumenta la tensión de puerta, que más tarde comprobaremos una vez implementado el modelo. Este trabajo es una generalización del modelo de MESFET desarrollado por nuestro grupo [2-3] con importantes modificaciones en las fuentes de corriente Ids e Igs para dar cuenta el efecto de la iluminación óptica. La expresión para la fuente de corriente Ids viene dada por (1) y cuyos parámetros más importantes se muestran en la tabla 1. El resto de las ecuaciones que componen el modelo han sido omitidas, puesto que han sido reportadas con anterioridad y una descripción detallada de las mismas puede encontrarse en [1].

$$I_{ds} = I_{dss} \cdot \left( I - \frac{V_{gi}}{V_t} \right)^{(E+K_e \cdot V_{gi})} \cdot \left( I + \frac{S_s \cdot V_{di}}{I_{dss}} \right) \cdot \tanh \left( \frac{S_l \cdot V_{di}}{I - K_g \cdot V_{gi}} \right) \cdot e^{\left( \frac{|V_{pf} - V_{gi}|^\delta}{\mu} \right)} \quad (1)$$

Parámetro	Significado físico	S.I.
$I_{dss}$	Corriente de drenador en saturación para $V_{gs}=0V$	A
$V_{to}$	Tensión de estrangulamiento de canal.	V
$E$	Parámetro que describe la variación de la corriente de drenador en saturación.	-
$K_e$	Parámetro que describe la variación de la corriente de drenador en saturación.	$V^{-1}$
$S_s$	Pendiente de la corriente de drenador en saturación	$\Omega^{-1}$
$K_g$	Parámetro que describe la zona lineal	$V^{-1}$
$S_l$	Pendiente de la corriente de drenador en la zona lineal para $V_{gi}=0V$ .	$\Omega^{-1}$
$V_{pf}$	Tensión a partir de la cual comienza a degradarse la trasconductancia	V
$\delta$	Parámetro de ajuste de la trasconductancia.	-
$\mu$	Parámetro de ajuste de la trasconductancia.	-

Tabla 1. Significado físico y unidades de los parámetros del modelo DICOM.

Las expresiones de la fuente de corriente mostrada en (1) utiliza como variables las tensiones anteriores internas del transistor  $V_{gi}$  y  $V_{di}$ , dichas tensiones son a su vez función de las tensiones aplicada externamente al dispositivo, estas son: la tensión gate-source y la tensión drain-source. La relación entre las mismas es la mostrada en (2) y (3).

$$V_{gi} = V_{gs} - I_d \cdot R_S \quad (1)$$

$$V_{di} = V_{ds} - I_d \cdot (R_d + R_S) \quad (2)$$

Es de rigor resaltar la importancia de obtener con exactitud el valor de las resistencias parásitas del transistor, por su implicación en el valor de las tensiones intrínsecas; de la precisión con la que se haya obtenido sus valores dependerá el funcionamiento del modelo. Se ha podido comprobar a lo largo de este trabajo que un pequeño cambio de estos valores resistivos condiciona bastante el comportamiento de la corriente de drenador del dispositivo.

#### IV. IMPLEMENTACIÓN DEL MODELO EN SPICE

Una vez seleccionadas las ecuaciones necesarias para el correcto funcionamiento del circuito propuesto se ha procedido a la integración de las mismas en del simulador *PSPICE* versión de estudiante de *ORCAD*. Dicho tarea se ha estructurado en tres partes las cuales serán comentadas a continuación. Es necesario resaltar que para que la implementación del modelo sea valida en las versiones del simulador *PSPICE*, no se ha utilizado los esquemáticos, escribiéndose todas las librerías como ficheros de texto.

En primer lugar, ha tenido que desarrollar una librería matemática denominada *MATH.LIB* la cual contiene una serie de sub-circuitos que implementan las funciones matemáticas mas habituales (suma, resta, tangente hiperbólica, exponenciación, etc.), cuya documentación se muestra en la figura 2.

```
*****
*                               LIBRERIA MATEMATICA : MATH.LIB
*                               *****
*-----
* Descripción de los SUBCKT Matematicos
*
* X1 1 2 3      ELEVA      V3=V1**V2
* X1 1 2 3      DIVIDE     V3=V1/V2
* X1 1 2 3      MULTIPLICA2 V3=V1*V2
* X1 1 2 3 4    MULTIPLICA3 V4=V1*V2*V3
* X1 1 2 3 4 5 MULTIPLICA4 V5=V1*V2*V3*V4
* X1 1 2 3      SUMA      V3=V1+V2
* X1 1 2 3      RESTA     V3=V1-V2
* X1 1 2 3      VTOI      Corriente de 2 a 3 = V1
* X1 1 2 3      VTOV      Tension de 2 a 3 = V1
* X1 1 2 3      GT        V3=1 si V1>V2
*                               V3=0 si V1<V2
*
* X1 1 2 3 4 5 COMP      V5=V1 si V3>V4
*                               V5=V2 si V3<V4
*-----
* X1 1 2      KMMUL      PARAMS:KMMUL= 2.0  V2=KMMUL*V1
* X1 1 2      P1         PARAMS:KAP1 KBP1  V2=A*V1+B
* X1 1 2      P2         PARAMS:KAP2 KBP2  V2=A*V1^2+B*V1+C
* X1 1 2      ELEVAK     PARAMS:KELEVAK=2  V2=V1**KELEVAK
* X1 1 2      EXPONEN    PARAMS:KEXPONEN=5  V2=EXP(KEXPONEN*V1)
* X1 1 2      LOGNEP     V2=LN(V1)
* X1 1 2      TANH       V2=TANH(V1)
* X1 1 2      COSH       V2=COSH(V1)
* X1 1 2      HFON       PARAMS:TAU=2.E-6  V2=V1*1/(1+S.Tau)
* X1 1 2      JFON       PARAMS:TAU=2.E-6  JFON=1-HFON
* X1 1 2      ABS        V2=ABS(V1)
*-----
```

Fig 2. Descripción de las funciones matemáticas contenidas en la librería de *PSPICE MATH.LIB* necesaria para simular dispositivos MESFET/HEMT de microondas en *PSPICE*. (los subcircuitos han sido omitidos aquí por la gran extensión de los mismos)

En segundo lugar, ha habido que definir cada uno de los sub-circuitos que componen el modelo completo, como el correspondiente a la fuente de corriente Ids (drain-source), el que modela la fuente de corriente Igs (gate-source), el de la fuente de corriente Igd (gate-drain) que describe el comportamiento del transistor en ruptura o “*breakdown*”, así como los circuitos necesarios para modelar la capacidad no lineal Cgs, que acompaña a la fuente Igs, y la capacidad lineal Cgd que acompaña a la fuente Igd. A modo de

ejemplo, en la Tabla 2 se muestra como se ha implementado la fuente de corriente no lineal  $I_{ds}$  en *PSPICE*, utilizando la ecuación (1).


Fuente $I_{ds}$	Implementación
	<p>Se ha nombrado el subcircuito como: <i>HEMTIDS</i>.            Posee 6 nudos, 4 de entrada y dos de salida.</p> <ul style="list-style-type: none"> <li>Nudos 1 y 2: Tensión de entrada <math>V_{gi}</math></li> <li>Nudos 3 y 4: Tensión de entrada <math>V_{di}</math></li> <li>Nudos 5 y 6: Corriente de salida</li> </ul> <p>La capacidad <math>C_{ds}</math> (lineal) no esta incluida en el subcircuito.</p>

Tabla 2. Implementación del conjunto fuente de corriente  $I_{ds}$  mas capacidad  $C_{ds}$  como sub-circuito en el simulador *PSPICE*.

La librería *HEMT\_GSD.LIB* contiene los sub-circuitos relacionados con las fuentes  $I_{gs}$  e  $I_{gd}$ , así como las capacidades  $C_{gs}$  y  $C_{gd}$

```

*****
* LIBRERIAS: TRANSISTOR HEMT *
*****
.SUBCKT HEMTIDS 1 2 3 4 5 6 PARAMS: KIDSS=81.952e-3
+ KVTO=-0.4223 GAMMAIDS=-0.075295 KE=1.3471 KKE=-0.4397
+ SS=0.003386 SL=2.3711 KG=-1.0811 KVPF=1.1723
+ DELTA=0.07497 + MU=1.1585
*Corriente en 5 y 6 como funcion de tension entre 1,2 Vgi
*y 3,4 Vdi
*IDS=IDSS*(1-Vgi/Vt)^(E+KE*Vgi)* (1+SS*Vdi/IDSS)*
*TANH(SL*Vdi/(1-Kg*Vgi))*EXP(-|Vpf-Vgi|^DELTA/MU)
R1 1 0 1E12
R2 2 0 1E12
EDIFF1 102 0 VALUE={V(1)-V(2)}
RDIFF1 102 0 1
R3 3 0 1E12
R4 4 0 1E12
EDIFF2 304 0 VALUE={V(3)-V(4)}
RDIFF2 304 0 1
EVT 305 0 VALUE={KVTO + GAMMAIDS*V(304)}
RVT 305 0 1
XDIV1 102 305 103 DIVIDE
VDC1 104 0 DC 1
RDC1 104 0 1
XRES1 104 103 105 RESTA
E1 106 0 VALUE={KE+KKE*V(102)}
*XP1 102 106 P1 PARAMS: KAP1={KKE} KBP1={KKE}
XPOT1 105 106 107 ELEVA
XKMUL1 107 108 KMUL PARAMS: KKMUL= {KIDSS}
E2 306 0 VALUE={SS*V(304)/KIDSS}
XSUM1 306 104 307 SUMA
XKMUL2 304 308 KMUL PARAMS: KKMUL= {SL}
XKMUL3 102 109 KMUL PARAMS: KKMUL= {KG}
XRES2 104 109 110 RESTA
XDIV2 308 110 309 DIVIDE
XTANH 309 310 TANH
E4 111 0 VALUE={KVPF-V(102)}
XABS 111 112 ABS
*E5 404 0 VALUE={PWR(V(403),DELTA)/MU}
XELEV 112 113 ELEVAK PARAMS: KELEVAK= {DELTA}
VDC2 114 0 {MU}
RDC2 114 0 1
XDIV3 113 114 115 DIVIDE
XEXP 115 116 EXPONEN PARAMS: KEXPONEN=-1.0
XMULT 108 307 310 116 400 MULTIPLICA4
XOUT 400 5 6 VTOI
.ENDS HEMTIDS

```

Fig 3. Librería de *PSPICE HEMT\_GSD.LIB* necesaria para simular el transistor MESFET/HEMT de microondas.

En tercer y último lugar se ha creado un fichero en lenguaje *PSPICE* que agrupa el conjunto de fuentes, capacidades y resistencias que forman el transistor y es el fichero que usualmente se invoca para proceder a su simulación. El contenido de dicho fichero se muestra en la figura 4. Aunque en este trabajo se muestra como introducir un modelo circuital de gran señal para el caso particular de un transistor FET de microondas, el procedimiento puede ser generalizado para simular cualquier tipo de dispositivo electrónico que presente una característica I-V no lineal.

TRANSISTOR HEMT - (H3S) DIMENSIONES:4X30um

```


*****
* CIRCUITO PARA LA SIMULACION TRANSISTOR HEMT *
*****
* LIBRERIAS *
*****
.LIB C:/PSPICE/MATH.LIB
.LIB C:/PSPICE/HEMT_GSD.LIB
*****
* DESCRIPCION DEL CIRCUITO *
*****
V1 1 0 DC 0.75
LG 1 2A 1E-10
R1 2A 2 6.6
XD1 2 5 DIODO_HEMT_GS PARAMS: INS=2E-13 ALPHA=14 CJO=0.1P
+ VBI=0.7 +M=0.5
XD2 2 4 DIODO_HEMT_GD PARAMS: INS=2E-15 ALPHA=13 CJO=0.1P
+ VBI=0.7 +M=0.5
V2 3 0 DC 6
LD 3 4A 1E-10
R2 4A 4 6.1
X1 2 5 4 5 4 5 HEMTIDS
CDS 4 5 15f
R3 5 5A 7.1
LS 5A 0 1E-10
*****
* ANALISIS EN CONTINUA *
*****
.DC V2 0 6 .5 V1 -0.7 1 0.1
*****
* PROBE+END *
*****
.PROBE
.END

```

Fig 4. Programa principal para simular el transistor MESFET/HEMT de microondas.

## V. RESULTADOS

A continuación se presentarán los resultados obtenidos mediante *PSPICE* para dos transistores PHEMT de microondas de la foundry PHILIPS uno de  $4*30\mu\text{m}$  y otro de  $6*150\mu\text{m}$  (potencia media). En las figuras 5(a) y 5(b) se muestran los resultados obtenidos en *PSPICE* mediante el post-procesador gráfico *PROBE* para el transistor PHEMT de  $4*30\mu\text{m}$ . En la figura 5(a) puede apreciarse la variación de  $I_{ds}$  frente a  $V_{ds}$  con  $V_{gs}$  como parámetro, y en la figura 5(b) se muestra la corriente  $I_{ds}$  frente a  $V_{gs}$  con  $V_{ds}$  como parámetro. En esta última gráfica puede apreciarse con claridad el fenómeno de compresión de la transconductancia típico de los transistores HEMT. En las figuras 6(a) y 6(b) se muestran los resultados obtenidos en *PSPICE* leídos directamente del fichero de salida *.OUT* de *PSPICE* mediante un pequeño programa realizado en *MATLAB* comparados con las medidas experimentales realizadas en el laboratorio para los transistores PHEMT de PHILIPS de  $4*30\mu\text{m}$  y  $6*150\mu\text{m}$ , respectivamente, tomándose los barridos en  $V_{gs}$  y  $V_{ds}$  en los mismos puntos que se habían medido experimentalmente en el laboratorio.






Fig 5. Resultados de la simulación con PSPICE para el transistor PHEMT de 4\*30um.

(a) Curvas características Ids frente Vds a Vgs constante.

(b) Curvas características Ids frente Vgs a Vds constante.

(Nótese el efecto de compresión de la transconductancia)

En la figura 6(a) para el dispositivo de 4\*30um el barrido en Vgs varia de -0.75V a +0.75V con paso de 0.25V, mientras que el barrido en Vds varia desde 0V a 4V con paso de 0.25V. En la figura 6(b) para el dispositivo de 6\*150um el barrido en Vgs va de -0.75V a +0.75V con paso de 0.25V, mientras que el barrido en Vds varia desde 0V a 6V con paso de 0.25V. En ambas figuras, se puede apreciar la perfecta concordancia entre los datos experimentales y los modelos desarrollados.

## VI. CONCLUSIONES

Se ha implementado un modelo circuital de gran señal en el simulador PSPICE, extendiendo la utilización del citado simulador a transistores de microondas. El método aquí presentado es generalizable a todo tipo de dispositivos no lineales activos o pasivos. Los resultados obtenidos concuerdan perfectamente con los valores experimentales, obteniéndose resultados similares a los obtenidos con simuladores de tipo balance armónico como Agilent ADS, incluyendo el fenómeno de compresión de la transconductancia. De esta forma utilizando una herramienta gratuita como es la versión de estudiante de PSPICE se pueden obtener con precisión las curvas I-V de transistores de microondas.

Como futura línea de trabajo se plantea el desarrollo de nuevas rutinas que permitan obtener los parámetros de scattering para este tipo de dispositivos con PSPICE, utilizando el modelo completo [1]. Otra línea de trabajo consiste en la introducción de los modelos electro-ópticos [5, 6, 7] de una forma análoga a como se ha introducido el modelo aquí presentado.

## AGRADECIMIENTOS

Los autores quieren agradecer al grupo de RF & Microondas de la Universidad de Cantabria, así como al proyecto NoE TARGET (IST program of the EU under contract IST-1-707893-NOE) las facilidades prestadas para la realización del presente trabajo.

## REFERENCIAS

- [1] A. Mediavilla, A. Tazón, J.L. García, T. Fernández, J.A. García, J.M. García, C. Navarro, J.M. Zamanillo. "Dynamic Properties and Modelling of Large Signal, Thermal, Optical and Intermodulation Effects in Microwave GaAs Devices", Invited paper on the IEEE MTT-S Workshop, Vol.1, Denver (USA). June 8-13 1997

- [2] J.M. Zamanillo, C. Navarro, C. Pérez-Vega, A. Mediavilla, and A. Tazón "Large Signal Model Predicts Dynamic Behavior of GaAs MESFET Under Optical Illumination.". Microwave and Optical Technology Letters. Vol. 29, No.1, pp 25-31. April 5 2001.
- [3] C. Navarro, J.M. Zamanillo, A. Mediavilla, A. Tazón, J.L. García, M. Lomer and J.M. López-Higuera. "An Accurate Photonic Capacitance Model for GaAs MESFET". IEEE Transactions on MTT, Vol 50, No. 4, pp 1193-1197, April 2002.
- [4] G. Massobrio, P. Antognetti "Semiconductor device modelling with SPICE". Mc Graw Hill 1993
- [5] J. M. Zamanillo, C. Navarro, C. Pérez-Vega, J. A. García, A. Mediavilla and A. Tazón. "New Large Signal Model of AlGaAs PHEMT and GaAs MESFET Under Optical Illumination". GaAs 2002 European Gallium Arsenide, Applications Symposium Proceedings, pp.121-124.Milan, Italy. Sept. 2002.
- [6] J.M. Zamanillo, J. Portilla, C. Navarro, C. Pérez-Vega and A. Mediavilla, "Optical Control of a GaAs MMIC Amplifier at S Band". GaAs 2003 European Gallium Arsenide, Applications Symposium Proceedings, pp.117-120, Munich, Germany. Oct. 2003
- [7] J.M. Zamanillo, C. Navarro, J. Sáiz-Ipiña, C. Pérez-Vega and A. Mediavilla. "New Large Signal Electrical Model of GaAs MESFET Under Optical Illumination". GaAs 2001 European Gallium Arsenide, Applications Symposium Proceedings, pp.167-170, London, Sept-2001.




Fig 5. Gráficas comparativas realizadas en MATLAB a partir de los resultados obtenidos en PSPICE y las medidas experimentales realizadas en el laboratorio para los transistores de 4\*30um y 6\*150um.

(a) Transistor PHEMT de PHILIPS de 4\*30um

(b) Transistor PHEMT de PHILIPS de 6\*150um.